Motivation
$global_clock is required to specify the global clock - the reference (or system) clock synchronizing the system transitions. To map the global clock in simulation the global clocking construct is provided. Ideally, the global clocking should specify the finest granularity clock in the entire design. However, many contemporary designs are multiclock, and may have more than 10 different clock domains, and building one reference clock would negatively affect the simulation performance.
In many cases is sufficient to define local reference clocks for different design blocks. This solution is accptable unless there is a need to write an assertion governed by the reference clock where signals from different clock domains are involved.
Suggested Solution
14.14 Global clocking
REPLACE
One clocking block may be declared as the global clocking for an entire elaborated SystemVerilog model.
The syntax for the global clocking declaration is as follows in Syntax 14-4.

clocking_declaration ::=                                                                                                                      // from A.6.11
...
| global clocking [ clocking_identifier ] clocking_event ; endclocking [ : clocking_identifier ]
Syntax 14-4—Global clocking syntax (excerpt from Annex A)
There shall be at most one global clocking declaration anywhere in an entire elaborated SystemVerilog model. It shall be an error if there is more than one such global clocking declaration. It shall be an error to place a global clocking declaration within a program block.
The system function $global_clock returns the event expression specified in the unique global clocking declaration. The function has no arguments. It shall be an error to invoke the $global_clock system function if there is no global clocking declaration in the elaborated SystemVerilog model. Otherwise, $global_clock may be invoked anywhere that a clocking event may be specified. 
The main purpose of global clocking is to specify which clocking event in simulation corresponds to the primary clock used in formal verification.
The following is an example of a global clocking declaration:
module top; 
    logic clk1, clk2;
    global clocking sys @(clk1 or clk2); endclocking
    // ...
endmodule
In this example, sys is declared as the global clocking event and is defined to occur if, and only if, there is a change of either of two signals, clk1 and clk2. Specification of the name sys in the global clocking declaration is optional since the global clocking event may be referenced by $global_clock from anywhere in the SystemVerilog model.
Any clocking_event may be specified in a global clocking declaration.
WITH
One clocking block may be declared as the global clocking for an entire elaborated SystemVerilog model.
A clocking block may be declared as the global clocking for the whole or a part of the elaborated SystemVerilog model. The main purpose of global clocking is to specify which clocking event in simulation corresponds to the primary system clock used in formal verification (see F.5.1). Such a specification may be done for a whole design, or separately for different subsystems in a design, when there are multiple clocks and building a single global clocking event is challenging.
MK> Remove “, when there are multiple clocks and building a single global clocking event is challenging.” No need to mention this 
The syntax for the global clocking declaration is as follows in Syntax 14-4.

clocking_declaration ::=                                                                                                                      // from A.6.11
...
| global clocking [ clocking_identifier ] clocking_event ; endclocking [ : clocking_identifier ]
Syntax 14-4—Global clocking syntax (excerpt from Annex A)
There shall be at most one global clocking declaration anywhere in an entire elaborated SystemVerilog model. It shall be an error if there is more than one such global clocking declaration. It shall be an error to place a global clocking declaration within a program block.
The system function $global_clock returns the event expression specified in the unique global clocking declaration. The function has no arguments. It shall be an error to invoke the $global_clock system function if there is no global clocking declaration in the elaborated SystemVerilog model. Otherwise, $global_clock may be invoked anywhere that a clocking event may be specified. 
Global clocking may be declared in a module, an interface, a checker, or a program. In a given module, interface, checker or a program a global clocking shall be declared at most once. It shall be an error if there is more than one global clocking declaration in a given module, interface, checker or program.
MK>We have same issue here related to generate blocks which we were discussing in the meeting. Is global clocking going to be allowed in generate blocks ? If yes, what about for generates etc. Also, will it only apply with the generate or apply to whole module ?
Any clocking_event may be specified in a global clocking declaration.
MK> Why is the above statement needed ? It will be decided by the BNF what is allowed ? 
Although more than one global clocking declaration may appear syntactically in a source description, only one global clocking declaration is in effect in (should be at) each point in the elaborated description. The reference to the event expression in the global clocking declaration in effect shall be done with $global_clock system function. The global clocking declaration in effect in (should be for ?) a specific $global_clock call site (site word is not needed) is determined using the following hierarchical lookup rules, which iteratively check the elaborated hierarchy of the design to find the global clocking declaration closest to the $global_clock call site.  
MK> These look up rules will also depend on if global clocking is allowed in generate blocks or not ?
a) Look for a global clocking declaration in the enclosing module, interface, checker or program scope. If found, the lookup terminates with the result being the event expression of that global clocking declaration. If not found and the current scope is the design element scope, the lookup terminates and shall result with an error. Otherwise, proceed to step b).
MK> Do you mean “design element scope” or top level module ? Any module, checker, program etc is a design element scope.
b) Look for a global clocking declaration in the parent module/interface/checker scope of the enclosing module/interface/checker/program instantiation. If found, the lookup terminates with the result being the event expression of that global clocking declaration. Otherwise, continue up the hierarchy until a global clocking declaration is found or the design element scope is reached. If no global clocking declaration is found and the design element scope is reached, the lookup terminates and shall result with an error.
MK> Again it should be ‘top level module’ in place of ‘design element scope’.
When $global_clock is used in a sequence declaration, a property declaration, or as an actual argument to a named sequence instance, a named property instance, or a checker instance, its call site shall be considered as if after the application of the rewriting algorithm defined in F.4.1, which flattens properties and sequences, and substitutes actual arguments to sequence, property and checker instances for their corresponding formal arguments. As a result, when a property or a sequence declaration containing a reference to $global_clock is instantiated in an assertion statement, the hierarchical lookup rules described above shall be applied from the place of the assertion statement appearance in the source description, not from the point of the sequence or the property declaration. Similarly, when $global_clock is used in an actual argument of a checker instance, the lookup rules shall be applied after the substitution of the actual argument in place of the corresponding argument inside the checker body.
MK> The first sentence of this paragraph is quite long and complicated. No need to describe re-writing algorithm here. Just mentioning it is enough. Consider the case of sequence being used as event outside of assertions. You can check some of the language from 2412. 
The main purpose of global clocking is to specify which clocking event in simulation corresponds to the primary clock used in formal verification.
The following is an example of a global clocking declaration:
module top; 
    logic clk1, clk2;
    global clocking sys @(clk1 or clk2); endclocking
    // ...
endmodule
MK> The above example was already there so it should not be blue.
In this example, sys is declared as the global clocking event and is defined to occur if, and only if, there is a change of either of two signals, clk1 and clk2. Specification of the name sys in the global clocking declaration is optional since the global clocking event may be referenced by $global_clock from anywhere in the SystemVerilog model.
MK> I think the part “from anywehre in the SystemVerilog model” should be removed as it is not true any more.
In the following example the elaborated hierarchy of the design contains two global clocking declarations. The call to $global_clock in top.susbsytem1.common resolves to the clock top.susbsystem1.subclk1. The call to $global_clock in top.susbsytem2.common resolves to the clock top.susbsystem2.subclk2.
module top; 
    subsystem1 sub1();
    subsystem2 sub2();
endmodule
module subsystem1;
    logic subclk1;
    global clocking sub_sys1 @(subclk1); endclocking

    common_sub common();
endmodule
module subsystem2;
    logic subclk2;
    global clocking sub_sys2 @(subclk2); endclocking

    common_sub common();
endmodule
module common_sub;
    always @($global_clock) begin
        // ...
    end
endmodule
In the following example the property p is declared in a module which contains a global clocking declaration. However, that global clocking declaration is not in effect where the property p is instantiated. Similarly to the previous example, the call to $global_clock in top.susbsytem1.checks resolves to the clock top.susbsystem1.subclk1, and the call to $global_clock in top.susbsytem2.checks resolves to  the clock top.susbsystem2.subclk2.
module top; 
    subsystem1 sub1();
    subsystem2 sub2();
endmodule
module subsystem1;
    logic subclk1, req, ack;
    global clocking sub_sys1 @(subclk1); endclocking

    common_checks checks(req, ack);
endmodule
module subsystem2;
    logic subclk2, req, ack;
    global clocking sub_sys2 @(subclk2); endclocking

    common_checks checks(req, ack);
endmodule
module another_module;
    logic another_clk;
    global clocking another_clocking @(another_clock); endclocking

    property p(req, ack);
        @($global_clock) req |=> ack;
    endproperty
endmodule
checker common_checks(logic req, logic ack);
    assert property another_module.p(req, ack);
endchecker
In the following example, $global_clock is used in a task. Since task calls do not carry substitution semantics, the call to $global_clock in the task another_module.t resolves to another_module.another_clk.
module top; 
    subsystem1 sub1();
    subsystem2 sub2();
endmodule
module subsystem1;
    logic subclk1, req, ack;
    global clocking sub_sys1 @(subclk1); endclocking

    always another_module.t(req, ack);
endmodule
module subsystem2;
    logic subclk2, req, ack;
    global clocking sub_sys2 @(subclk2); endclocking

    always another_module.t(req, ack);
endmodule
module another_module;
    logic another_clk;
    global clocking another_clocking @(another_clk); endclocking

    task t(input req, input ack);
        @($global_clock);
        // ...
    endtask
endmodule
The following example demonstrates the usage of $global_clock in checker arguments. The resolution of the calls to $global_clock is performed after the substitution of the actual checker arguments in place of the corresponding formal arguments, and after the flattening of the properties in the assertion statements, therefore all calls to $global_clock in this example refer to top.check.checker_clk.
module top; 
    logic a, b, c, clk;
    global clocking top_clocking @(clk); endclocking

    property p1(req, ack);
        @($global_clock) req |=> ack;
    endproperty

    property p2(req, ack, interrupt);
        @($global_clock) accept_on(interrupt) p1(req, ack);
    endproperty

    my_checker check(
        p2(a, b, c),
        @$global_clock a[*1:$] ##1 b);
endmodule
checker my_checker(property p, sequence s);
    logic checker_clk;
    global clocking checker_clocking @(checker_clk); endclocking

    assert property (p);
    cover property (s);
endchecker
Any clocking_event may be specified in a global clocking declaration.
NOTE—This is an area of backwards-incompatibility between this standard and 14.14 of IEEE Std 1800-2009. In the 2009 definition only one global clocking declaration was allowed in the design description, it could be specified in any module, interface or checker and apply everywhere in the description. A design conforming to IEEE Std 1800-2009 could have a global clocking declaration defined in a non-top module and use $global_clock out of the subhierarchy of that non-top module. Such a design shall not be conforming to this standard.
16.5 Concurrent assertions overview
REPLACE

A reference to $global_clock (see 14.14) is understood to be a reference to the clocking_event defined in
the global clocking declaration. The global clock behaves just as any other clocking event. In formal
verification, however, $global_clock has additional significance, as it is considered to be the primary
system clock (see F.5.1). Thus, in the following example:
WITH

A reference to $global_clock (see 14.14) is understood to be a reference to a the clocking_event defined in
a the global clocking declaration. A The global clock behaves just as any other clocking event. In formal
verification, however, $global_clock has additional significance, as it is considered to be the primary
system clock (see F.5.1). Thus, in the following example:
